
Summary

The following software explains how an array of
structures are made.  It also shows how to convert a
date into a string.

APPLICATION
NOTE

MI-AN-019

Arrays of Structures and
Date to String Conversion

\ MI-AN-019     Arrays of Structures and Date to String Conversion
\ The following code demonstrates how to make an array of structures and

convert
\ the date returned by READ.WATCH into a string.

DECIMAL
STRUCTURE.BEGIN: READING

                 REAL-> +Voltage
                 BYTE-> +Year
                 BYTE-> +Month
                 BYTE-> +Day

STRUCTURE.END

ARRAY: READINGS
10     CONSTANT NUM.READINGS

\ Init array READINGS as a single dimension of NUM.READINGS structures.
: INIT.READINGS ( -- )

  NUM.READINGS 1 READING ' READINGS DIMENSIONED \ Dim 1 dimensional array
  ' READINGS ZERO.ARRAY \ Init array to all zeros

;

\ Reads the real time clock and drops everything but the date, month and yr
: DATE ( -- date\month\year )

  READ.WATCH
  >R >R >R 5 NDROP R> R> R>

;

\ Stores the voltage on the stack at the given READING structure address.
\ The date is stored with the voltage.
: STORE.VOLTAGE ( f.voltage\reading.xaddr -- )

  LOCALS{ x&reading }
  x&reading +Voltage F! \ Store the voltage reading
  DATE \ Get the current date
  x&reading +Year C! \ Store the year
  x&reading +Month C! \ Store the month
  x&reading +Day C! \ Store the date

;

\ DATE>$ creates a string at PAD with this format mm/dd/yy. To
\ test this word type: 24 8 94 DATE>$ TYPE. NOTE: The result must
\ immediately be TYPEd or cmoved since PAD is used as a temporary
\ buffer for a variety of other functions.
: DATE>$ ( day\month\year -- string.xaddr\cnt )

  ROT SWAP \ rearrange stack = month\day\year --
  BASE @ >R \ save the current base
  DECIMAL \ make sure conversion is done in decimal
  <# S>D  #  #S  2DROP \ make year a double and convert

ASCII  /  HOLD \ insert a slash
 S>D  #  #S  2DROP \ make day a double and convert

ASCII  /  HOLD \ insert a slash
S>D  #  #S \ make month a double and convert

  #> \ push string xaddr and cnt, cnt addr = PAD
  R> BASE ! \ restore original base

;

\ TYPE.READING given a reading structure will type the voltage associated
\ with a given date.
: TYPE.READING ( reading.xaddr -- )

  LOCALS{ x&reading }
  x&reading +Voltage F@
  CR  ."  V = "  F.   ."  was read on "
  x&reading +Day C@
  x&reading +Month C@
  x&reading +Year C@
  DATE>$ TYPE

;

\ Demonstrate that this code works...
: TEST.READINGS ( -- )

  INIT.READINGS
  NUM.READINGS 0
  DO
    I FLOT PI F* \ Push a simulated voltage = PI * I
    I READINGS STORE.VOLTAGE
  LOOP
  NUM.READINGS 0
  DO
    I READINGS TYPE.READING
  LOOP

;

